

10.5 FINITION LATÉRALE OCM (cycle 274, DIN/ISO : G274, option 167)

Déroulement du cycle

Le cycle 274 **FINITION LATÉRALE OCM** réalise la finition de la surépaisseur programmée dans le cycle 271. Ce cycle peut être exécuté aussi bien en avalant qu'en opposition.

Avant d'appeler le cycle 274, vous devez d'abord programmer d'autres cycles :

- **CONTOUR DEF**, sinon le cycle 14 **CONTOUR**
 - Cycle 271 **DONNEES CONTOUR OCM**
 - Au besoin, le cycle 272 **EBAUCHE OCM**
 - Cycle 273 **PROF. FINITION OCM**
- 1 La commande positionne l'outil au point de départ de la position d'approche, au-dessus de la pièce. Cette position dans le plan est obtenu à partir d'une trajectoire circulaire tangentielle sur laquelle la CN déplace l'outil.
 - 2 La commande amène ensuite l'outil à la première profondeur de passe, avec l'avance définie pour la passe en profondeur.
 - 3 La CN approche et quitte le contour selon un arc hélicoïdal tangentiel, jusqu'à la fin de la finition de l'ensemble du contour. L'opération de finition s'effectue séparément pour chaque partie de contour.
 - 4 L'outil retourne ensuite à la hauteur de sécurité, dans l'axe d'outil.

Vous pouvez aussi utiliser le cycle 274 pour le fraisage de contours. Procédez comme suit :

- ▶ Définir le contour à fraiser comme îlot individuel (sans limitation de poche)
- ▶ Programmer dans le cycle 271 la surépaisseur de finition (**Q368**) de manière à ce qu'elle soit supérieure à la somme de la surépaisseur de finition **Q14** et du rayon de l'outil utilisé

Attention lors de la programmation !



Ce cycle ne peut être exécuté qu'en mode **FUNCTION MODE MILL**.

La surépaisseur latérale **Q14** reste après l'opération de finition. Cette surépaisseur doit toutefois être inférieure à la surépaisseur dans le cycle 271.

La commande détermine automatiquement le point de départ de la finition. Le point de départ dépend de l'espace disponible sur le contour et de la surépaisseur programmée dans le cycle 271.



Vous devez définir un outil d'évidement dans le paramètre de cycle **Q438**, sinon la CN émet un message d'erreur.

Vous pouvez exécuter le cycle avec un outil de rectification.

Paramètres du cycle



- ▶ **Q338 Passe de finition?** (en incrémental) : cote de la passe de finition de l'outil sur l'axe de la broche.
Q338=0: finition en une seule passe. Plage de programmation : 0 à 99999,9999
- ▶ **Q385 Avance de finition?** : vitesse de déplacement de l'outil lors de la finition latérale, en mm/min. Plage de programmation : 0 à 99999,999 sinon **FAUTO, FU, FZ**
- ▶ **Q253 Avance de pré-positionnement?** : vitesse de déplacement de l'outil lors de l'approche de la position de départ. Cette avance est utilisée sous la surface de coordonnées mais hors du matériau défini. En mm/min. Plage de programmation : 0 à 99999,9999 ou **FMAX, FAUTO, PREDEF**
- ▶ **Q200 Distance d'approche?** (en incrémental) : distance entre l'arête inférieure de l'outil et la surface de la pièce. Plage de programmation : 0 à 99999,9999
- ▶ **Q14 Surepaisseur finition latérale?** (en incrémental) : la surépaisseur latérale **Q14** reste après l'opération de finition. (Cette surépaisseur doit toutefois être inférieure à la surépaisseur dans le cycle 271.) Plage de programmation : -99999,9999 à 99999,9999
- ▶ **Q438 Numéro/Nom outil d'évidement?** **Q438** ou **Q5438** : numéro ou nom de l'outil avec lequel la commande a effectué l'évidement de la poche de contour. Vous avez la possibilité de reprendre directement, par softkey, l'outil de pré-évidement du tableau d'outils. Vous pouvez en outre utiliser la softkey **Nom d'outil** pour indiquer le nom d'outil. Lorsque vous quittez le champ de saisie, la commande insère automatiquement le premier guillemet. Plage de programmation pour les valeurs numériques : -1 à +32767,9
Q438=-1: Le dernier outil utilisé est considéré comme l'outil d'évidement (comportement par défaut).
- ▶ **Q351 Sens? en aval.=+1, en oppos.=-1** : type de fraisage. Le sens de rotation de la broche est pris en compte :
+1 = fraisage en avalant
-1 = fraisage en opposition
PREDEF : la CN utilise la valeur de la séquence GLOBAL DEF. (Si vous indiquez la valeur 0, l'usinage se fera en avalant.)



Exemple

61 CYCL DEF 274 FINITION LATÉR. OCM	
Q338=+0	;PASSE DE FINITION
Q385=+500	;AVANCE DE FINITION
Q253=+750	;AVANCE PRE-POSIT.
Q200=+2	;DISTANCE D'APPROCHE
Q14=+0	;SUREPAIS. LATÉRALE
Q438=-1	;NUMÉRO/NOM OUTIL D'ÉVIDEMENT?
Q351=+1	;MODE FRAISAGE


10.6 Exemples de programmation

Exemple : Poche ouverte et reprise d'évidement avec des cycles OCM

Le programme CN suivant fait appel aux cycles OCM. Une poche ouverte est programmée. Cela se fait via une limite et un îlot.

Déroulement du programme

- Appeler la fraise d'ébauche
- Définir **CONTOUR DEF**
- Définition du cycle 271
- Définir et appeler le cycle 272
- Appeler la fraise de finition
- Définir et appeler le cycle 273
- Définir et appeler le cycle 274



0 BEGIN PGM OCM_POCKET MM	
1 BLK FORM 0.1 Z X+0 Y+0 Z-30	Définition de la pièce brute
2 BLK FORM 0.2 X+100 Y+100 Z+0	
3 TOOL CALL "MILL_D20" Z S8000 F1500	Appel de l'outil, diamètre 20
4 M3	
5 L Z+250 R0 FMAX	
6 L X+0 Y+0 R0 FMAX	
7 CONTOUR DEF P1 = LBL 1 I2 = LBL 2	
8 CYCL DEF 271 DONNEES CONTOUR OCM	Définition des paramètres d'usinage
Q203=+0 ;COORD. SURFACE PIECE	
Q201=-10 ;PROFONDEUR	
Q368=+0.5 ;SUREPAIS. LATERALE	
Q369=+0.5 ;SUREP. DE PROFONDEUR	
Q260=+100 ;SICHERE HOEHE	
Q578=+0.2 ;FACTEUR COIN INTERIEUR	
Q569=+1 ;LIMITE OUVERTE	
9 CYCL DEF 272 EBAUCHE OCM	Définir le cycle d'ébauche
Q202=+5 ;PROFONDEUR DE PASSE	
Q370=+0.4 ;FACTEUR RECOUVREMENT	
Q207= AUTO ;AVANCE FRAISAGE	
Q568=+0.6 ;FACTEUR DE PLONGEE	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
Q438=+0 ;OUTIL EVIDEMENT	
Q577=+0.2 ;FACT. RAYON D'APPROCHE	
Q351=+1 ;MODE FRAISAGE	
10 CYCL CALL	Appel du cycle
11 TOOL CALL "MILL_D8" Z S8000 F1500	Appel de l'outil, diamètre 8
12 M3	

13 L Z+250 R0 FMAX	
14 L X+0 Y+0 R0 FMAX	
15 CYCL DEF 272 EBAUCHE OCM	Définir le cycle d'ébauche
Q202=+5 ;PROFONDEUR DE PASSE	
Q370=+0.4 ;FACTEUR RECOUVREMENT	
Q207= AUTO ;AVANCE FRAISAGE	
Q568=+0.6 ;FACTEUR DE PLONGEE	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
QS438="MILL_D20" ;OUTIL EVIDEMENT	
Q577=+0.2 ;FACT. RAYON D'APPROCHE	
Q351=+1 ;MODE FRAISAGE	
16 CYCL CALL	Appel du cycle
17 TOOL CALL "MILL_D6_FINISH" Z S10000 F2000	Appel de l'outil, diamètre 6
18 M3	
19 L Z+250 R0 FMAX	
20 L X+0 Y+0 R0 FMAX	
21 CYCL DEF 273 PROF. FINITION OCM	Définir la profondeur du cycle d'ébauche
Q370=+0.8 ;FACTEUR RECOUVREMENT	
Q385= AUTO ;AVANCE DE FINITION	
Q568=+0.3 ;FACTEUR DE PLONGEE	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
Q438=-1 ;OUTIL EVIDEMENT	
22 CYCL CALL	Appel du cycle
23 CYCL DEF 274 FINITION LATER. OCM	Définir le cycle de finition latérale
Q338=+0 ;PASSE DE FINITION	
Q385= AUTO ;AVANCE DE FINITION	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
Q14=+0 ;SUREPAIS. LATERALE	
QS438=-1 ;OUTIL EVIDEMENT	
Q351=+1 ;MODE FRAISAGE	
24 CYCL CALL	Appel du cycle
25 M30	Fin du programme
26 LBL 1	Sous-programme de contour 1
27 L X+0 Y+0	
28 L X+100	
29 L Y+100	
30 L X+0	
31 L Y+0	
32 LBL 0	
33 LBL 2	Sous-programme de contour 2


34 L X+0 Y+0	
35 L X+100	
36 L Y+100	
37 L X+70	
38 L Y+70	
39 RND R5	
40 L X+30	
41 L Y+100	
42 RND R5	
43 L X+0	
44 L Y+0	
45 LBL 0	
46 END PGM OCM_POCKET MM	

Exemple : Différentes profondeurs avec des cycles OCM

Le programme CN suivant fait appel aux cycles OCM. Une poche est définie, ainsi que deux îlots, à des hauteurs variées.

Déroulement du programme

- Appeler la fraise d'ébauche
- Définir **CONTOUR DEF**
- Définition du cycle 271
- Définir et appeler le cycle 272
- Appeler la fraise de finition
- Définir et appeler le cycle 273
- Définir et appeler le cycle 274



0 BEGIN PGM OCM_DEPTH MM	
1 BLK FORM 0.1 Z X-50 Y-50 Z-30	Définition de la pièce brute
2 BLK FORM 0.2 X+50 Y+50 Z+0	
3 TOOL CALL "MILL_D10" Z S8000 F1500	Appeler l'outil de diamètre D10
4 L Z+250 R0 FMAX M3	
5 L X+0 Y+0 R0 FMAX	
6 CONTOUR DEF P1 = LBL 1 I2 = LBL 2 I3 = LBL 3 DEPTH5	
7 CYCL DEF 271 DONNEES CONTOUR OCM	Définition des paramètres d'usinage
Q203=+0 ;COORD. SURFACE PIECE	
Q201=-15 ;PROFONDEUR	
Q368=+0.5 ;SUREPAIS. LATERALE	
Q369=+0.5 ;SUREP. DE PROFONDEUR	
Q260=+100 ;SICHERE HOEHE	
Q578=+0.2 ;FACTEUR COIN INTERIEUR	
Q569=+0 ;LIMITE OUVERTE	
8 CYCL DEF 272 EBAUCHE OCM	Définir le cycle d'ébauche
Q202=+5 ;PROFONDEUR DE PASSE	
Q370=+0.4 ;FACTEUR RECOUVREMENT	
Q207= AUTO ;AVANCE FRAISAGE	
Q568=+0.6 ;FACTEUR DE PLONGEE	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
Q438=+0 ;OUTIL EVIDEMENT	
Q577=+0.2 ;FACT. RAYON D'APPROCHE	
Q351=+1 ;MODE FRAISAGE	
9 CYCL CALL	Appel du cycle
10 TOOL CALL "MILL_D6_FINISH" Z S10000 F2000	Appeler l'outil de diamètre D6
11 M3	
12 L Z+250 R0 FMAX	
13 L X+0 Y+0 R0 FMAX	

14 CYCL DEF 273 PROF. FINITION OCM	Définir la profondeur du cycle d'ébauche
Q370=+0.8 ;FACTEUR RECOUVREMENT	
Q385= AUTO ;AVANCE DE FINITION	
Q568=+0.3 ;FACTEUR DE PLONGEE	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
Q438=-1 ;OUTIL EVIDEMENT	
15 CYCL CALL	Appel du cycle
16 CYCL DEF 274 FINITION LATER. OCM	Définir le cycle de finition latérale
Q338=+0 ;PASSE DE FINITION	
Q385= AUTO ;AVANCE DE FINITION	
Q253=+750 ;AVANCE PRE-POSIT.	
Q200=+2 ;DISTANCE D'APPROCHE	
Q14=+0 ;SUREPAIS. LATERALE	
QS438="MILL_D10" ;OUTIL EVIDEMENT	
Q351=+1 ;MODE FRAISAGE	
17 CYCL CALL	Appel du cycle
18 M30	Fin du programme
19 LBL 1	Sous-programme de contour 1
20 L X-40 Y-40	
21 L X+40	
22 L Y+40	
23 L X-40	
24 L Y-40	
25 LBL 0	
26 LBL 2	Sous-programme de contour 2
27 L X-10 Y-10	
28 L X+10	
29 L Y+10	
30 L X-10	
31 L Y-10	
32 LBL 0	
33 LBL 3	Sous-programme de contour 3
34 L X-20 Y-20	
35 L Y+20	
36 L X+20	
37 L Y-20	
38 L X-20	
39 LBL 0	
40 END PGM OCM_DEPTH MM	

11

**Cycles d'usinage :
corps d'un cylindre**

11.1 Principes de base

Résumé des cycles sur corps d'un cylindre

Softkey	Cycle	Page
	27 CORPS D'UN CYLINDRE	329
	28 CORPS D'UN CYLINDRE Rainurage	332
	29 CORPS D'UN CYLINDRE Fraisage d'un ilot oblong	337
	39 CORPS D'UN CYLINDRE Fraisage d'un contour extérieur	340